Proceedings of the 1983
Winter Simulation Conference
S. Roberts, J. Banks, B. Schmeiser (eds.)

MODELLING CONSTRUCTS OF SIMSCRIPT II.5®

Edward C. Russell

CACI
12011 San Vicente Boulevard
Los Angeles, CA 90049

This tutorial will present the highlights of the SIMSCRIPT II.5 approach to building discrete event simulation models. The approach will be to construct a small example problem, implement the program in SIMSCRIPT II.5, and then to display the modularity which is possible with SIMSCRIPT by adding several "real-world" complexities to the model.

1. BACKGROUND

Simulation languages are (or should be) more than extensions of general-purpose programming languages designed to ease the burden of programming simulation problems. The influence of a good simulation language should be felt during the specification and model design stages of simulation as well as during computer implementation. If the "world-view" of the simulation language is well understood, and if that world-view is appropriate for a given problem, then the language should aid immeasurably in reducing the effort (and consequently elapsed time) in transforming model from concept to realization.

2. SIMSCRIPT II.5 WORLD-VIEW

SIMSCRIPT is a discrete-event language (Figure 1.) Actions are modelled in terms of events. Sequences of events describing actions of a single object (or entity) are modelled as processes. Many important relationships are described dynamically in terms of entities-attributes-sets. This very powerful data structuring is one of the unique features of the language. The implementation in SIMSCRIPT of the classical simulation problems, such as small queueing models or job-shop simulations, have been described elsewhere (see References).

3. SIMSCRIPT II.5 IMPLEMENTATIONS

SIMSCRIPT II.5 has been implemented on seven different manufacturer's computers including IBM, Honeywell, CDC, Univac, VAX, NCR, and PR1ME. A major emphasis is to achieve portability of models from one machine to another since the type of models typically written in SIMSCRIPT have a fairly long lifetime and are often shared among a community of users with different computers.

4. SIMSCRIPT II.5 AVAILABILITY

SIMSCRIPT II.5 is the proprietary product of CACI and is available from them for lease or purchase. SIMSCRIPT is also available on Boeing Computer Services, Control Data-Cybernet, Service Bureau Corporation, and United Computing Services.

A university program is supported by CACI in which SIMSCRIPT is supplied to educational institutions for the cost of distribution.

5. THE TUTORIAL EXAMPLE

The tutorial is built around a simple queueing problem first presented in (SCHRIBER) and later revised in (FISHMAN 1978). An African Port consists of three docks serviced by a single tug (Figure 2). Ships which arrive to be loaded must use the tug to enter and leave the docks. The ships arrive according to an exponential distribution of inter-arrival times with a mean of eleven hours. The ships are of three sizes distributed as follows:

25% have a mean loading time of 18 hours
55% have a mean loading time of 24 hours
20% have a mean loading time of 36 hours

CH1953-9/83/0000-0151 $01.00 © 1983 IEEE

The times are all exponentially distributed about these means. The time for moving a ship into or out of a dock is also exponentially distributed with a mean of one hour. Ships queue for the tug and/or the dock on a first-come-first-served basis.

The first task is to model this existing situation and measure the utilization factors for the tug and the dock. In addition, the congestion in the harbor (queueing statistics for the tug and dock) are to be measured. Finally, statistics on the in-port time for the ships should be collected.

After this simple model is developed, the next phase is to superimpose a new category of tanker. These new tankers belong to a fixed size fleet. They have a mean loading time of 21 hours and require the same tug and dock services as the other tankers. When they leave the port, they make a round-trip to their destination in a mean time of ten days (again exponentially distributed). The same statistics should be reported as above in order to determine the impact of the new tankers on the port.

The next complication to be added to the model is the effect of external interruptions such as storms and smog. A storm serves to delay the arrival of ships and to detain them when ready to depart. Smog alerts cause the unloading of ships to be interrupted until the smog alert is lifted. Since storms and smog are fairly rare occurrences, these phenomena are to be represented as they actually occurred over some historical period.

Finally, in order to verify the correct execution of the model, the SIMSCRIPT tools for debugging will be added to the model.

The tutorial shall be comprised of a detailed walkthrough of the original model and a discussion of the evolution to the completed model in the following steps:

1) original simple model with only old tankers (Figure 3);

2) additions for statistical output (Figure 4);

3) additions for the new tankers; (Figure 5);

4) additions for storms and smog; (Figure 6);

5) additions for debug (Figure 7).

REFERENCES

Fishman, G.S., Principles of Discrete Event Simulation, Wiley-Interscience, New York, 1978.

Kiviat, P.J., H.M. Markowitz, and R. Villanueva, SIMSCRIPT II.5 Programming Language (edited by Alasdar Mullarney), CACI, Los Angeles, 1983.

Russell, E.C. and J.S. Annino, A Quick Look at SIMSCRIPT II.5, CACI, Los Angeles, 1983.

Russell, E.C., Building Simulation Models with SIMSCRIPT II.5, CACI, Los Angeles, 1983.

Schriber, T.J., Simulation Using GPSS, Wiley, New York, 1974.



FIGURE 1
THE TIMING ROUTINE



FIGURE 2
EXAMPLE PROBLEM

```
"                SIMSCRIPT II.5 TUTORIAL MODEL
"                    AN AFRICAN PORT STUDY
PREAMBLE
    PROCESSES INCLUDE GENERATOR AND SHIP
    RESOURCES INCLUDE DOCK AND TUG
    THE SYSTEM HAS A LOADING.TIME RANDOM STEP VARIABLE
    DEFINE LOADING.TIME AS A REAL VARIABLE
    DEFINE .END.OF.SIMULATION TO MEAN REAL TIME.V >= 365.
END "PREAMBLE

MAIN
    READ LOADING.TIME
    CREATE EVERY DOCK(1)
    LET U.DOCK(1) = 3
    CREATE EVERY TUG(1)
    LET U.TUG(1) = 1
    ACTIVATE A GENERATOR NOW
    START SIMULATION
END "MAIN

INPUT DATA:
    0.25  18  0.55  24  0.2  36  *


PROCESS GENERATOR
    UNTIL .END.OF.SIMULATION,
    DO
        ACTIVATE A SHIP NOW
        WAIT EXPONENTIAL.F(11.0, 2) HOURS
    LOOP
    STOP
END "PROCESS GENERATOR

PROCESS SHIP
        REQUEST 1 DOCK(1)
        REQUEST 1 TUG(1)
        WAIT EXPONENTIAL.F(1.0,3) HOURS
        RELINQUISH 1 TUG(1)
        WORK EXPONENTIAL.F(LOADING.TIME, 4) HOURS
        REQUEST 1 TUG(1)
        WAIT EXPONENTIAL.F(1.0,5) HOURS
        RELINQUISH 1 TUG(1)
        RELINQUISH 1 DOCK(1)
END "PROCESS SHIP
```

FIGURE 3

```
"                SIMSCRIPT II.5 TUTORIAL MODEL
"                    AN AFRICAN PORT STUDY
PREAMBLE
    PROCESSES INCLUDE GENERATOR AND SHIP
    RESOURCES INCLUDE DOCK AND TUG
    THE SYSTEM HAS A LOADING.TIME RANDOM STEP VARIABLE
    DEFINE LOADING.TIME AS A REAL VARIABLE
    ACCUMULATE AVG.DOCK.QUEUE AS THE AVERAGE OF N.Q.DOCK
    ACCUMULATE DOCK.UTILIZATION AS THE AVERAGE OF N.X.DOCK
    ACCUMULATE TUG.UTILIZATION AS THE AVERAGE OF N,X.TUG
    ACCUMULATE AVG.TUG.QUEUE AS THE AVERAGE OF N.Q.TUG
    TALLY AVG.IN.PORT.TIME AS THE AVERAGE
        AND NO.OF.SHIPS.SERVED AS THE NUMBER
        AND MAX.IN.PORT.TIME AS THE MAXIMUM OF IN.PORT.TIME
    DEFINE IN.PORT.TIME AS A REAL VARIABLE
    DEFINE .END.OF.SIMULATION TO MEAN TIME.V >= 365.
END "PREAMBLE


MAIN
    READ TYPE.DISTRIBUTION
    PRINT 1 LINE THUS
                AFRICAN PORT TANKER STUDY
    SKIP 1 LINE
    CREATE EVERY DOCK(1)
    LET U.DOCK(1) = 3
    CREATE EVERY TUG(1)
    LET U.TUG(1) = 1
    ACTIVATE A GENERATOR NOW
    START SIMULATION
END "MAIN


PROCESS GENERATOR
    UNTIL .END.OF.SIMULATION,
    DO
        ACTIVATE A SHIP NOW
        WAIT EXPONENTIAL.F(11.0, 2) HOURS
    LOOP
    SKIP 2 LINES
    PRINT 11 LINES WITH TIME.V, NO.OF.SHIPS,SERVED,
        AVG.IN..PORT.TIME, MAX.IN.PORT.TIME,
        DOCK.UTILIZATION(1)/3, AVG.DOCK.QUEUE(1),
        TUG.UTILIZATION(1) AND AVG.TUG.QUEUE(1) THUS
                AFTER **.** DAYS

          * SHIPS HAVE BEEN LOADED.
        THE AVERAGE IN-PORT TIME FOR A SHIP WAS  *.** HOURS.
        THE MAXIMUM IN-PORT TIME FOR A SHIP WAS  *.** HOURS.

        THE DOCK UTILIZATION WAS  *.** (X 100% )
        THE AVERAGE QUEUE FOR THE DOCK WAS  *.** SHIPS

        THE TUG UTILIZATION WAS  *.** (X 100% )
        THE AVERAGE QUEUE FOR THE TUG WAS  *.** SHIPS
    STOP
END "PROCESS GENERATOR


PROCESS SHIP
    DEFINE ARRIVAL.TIME AS A REAL VARIABLE
        LET ARRIVAL.TIME = TIME.V
        REQUEST 1 DOCK(1)
        REQUEST 1 TUG(1)
        WAIT EXPONENTIAL.F(1.0,3) HOURS
        RELINQUISH 1 TUG(1)
        WORK EXPONENTIAL.F(LOADING.TIME,4) HOURS
        REQUEST 1 TUG(1)
        WAIT EXPONENTIAL.F(1.0,5) HOURS
        RELINQUISH 1 TUG(1)
        RELINQUISH 1 DOCK(1)
        LET IN.PORT.TIME = (TIME.V - ARRIVAL.TIME) * HOURS.V
END "PROCESS SHIP
```

```
                AFRICAN PORT TANKER STUDY

                    AFTER 365.02 DAYS

803 SHIPS HAVE BEEN LOADED.
  THE AVERAGE IN-PORT TIME FOR A SHIP WAS  45.24 HOURS.
  THE MAXIMUM IN-PORT TIME FOR A SHIP WAS 206.28 HOURS.

  THE DOCK UTILIZATION WAS   .77 (X 100 % )
  THE AVERAGE QUEUE FOR THE DOCK WAS  1.82 SHIPS

  THE TUG UTILIZATION WAS   .19 (X 100 % )
  THE AVERAGE QUEUE FOR THE TUG WAS   .03 SHIPS
```

FIGURE 4

```
"                SIMSCRIPT II.5 TUTORIAL MODEL
"                    AN AFRICAN PORT STUDY

PREAMBLE
    PROCESSES INCLUDE GENERATOR AND SHIP
        EVERY SHIP HAS A TYPE
        DEFINE TYPE AS AN ALPHA VARIABLE
    RESOURCES INCLUDE DOCK AND TUG
    DEFINE NO.NEW.SHIPS AS AN INTEGER VARIABLE
    THE SYSTEM HAS A LOADING.TIME RANDOM STEP VARIABLE
    DEFINE LOADING.TIME AS A REAL VARIABLE
    ACCUMULATE AVG.DOCK.QUEUE AS THE AVERAGE OF N.Q.DOCK
    ACCUMULATE DOCK.UTILIZATION AS THE AVERAGE OF N.X.DOCK
    ACCUMULATE TUG.UTILIZATION AS THE AVERAGE OF N.X.TUG
    ACCUMULATE AVG.TUG.QUEUE AS THE AVERAGE OF N.Q.TUG
    TALLY AVG.IN.PORT.TIME AS THE AVERAGE
        AND NO.OF.SHIPS.SERVED AS THE NUMBER
        AND MAX.IN.PORT.TIME AS THE MAXIMUM OF IN.PORT.TIME
    DEFINE IN.PORT.TIME AS A REAL VARIABLE
    DEFINE .END.OF.SIMULATION TO MEAN TIME.V >= 365.
END "PREAMBLE


MAIN
    READ LOADING.TIME
    PRINT 1 LINE THUS
                AFRICAN PORT TANKER STUDY
    SKIP 1 LINE
    READ NO.NEW.SHIPS
    PRINT 1 LINE WITH NO.NEW.SHIPS THUS
                NO. OF NEWS SHIPS IS *.
    SKIP 1 LINE
    CREATE EVERY DOCK(1)
    LET U.DOCK(1) = 3
    CREATE EVERY TUG(1)
    LET U.TUG(1) = 1
    ACTIVATE A GENERATOR NOW
    START SIMULATION
END "MAIN

INPUT DATA:
    5
    0.25  18  0.55  24  0.2  36  *


PROCESS GENERATOR
    DEFINE I AS AN INTEGER VARIABLE
    FOR I = 1 TO NO.NEW.SHIPS,
    DO
        ACTIVATE A SHIP IN 2 * I DAYS
        LET TYPE(SHIP) = "NEW"
    LOOP
    UNTIL .END.OF.SIMULATION,
    DO
        ACTIVATE A SHIP NOW
        LET TYPE(SHIP) = "OLD"
        WAIT EXPONENTIAL.F(11.0, 2) HOURS
    LOOP
    SKIP 2 LINES
    PRINT 11 LINES WITH TIME.V, NO.OF.SHIPS.SERVED,
        AVG.IN..PORT.TIME, MAX.IN.PORT.TIME,
        DOCK.UTILIZATION(1)/3, AVG.DOCK.QUEUE(1),
        TUG.UTILIZATION(1) AND AVG.TUG.QUEUE(1) THUS
                AFTER  **.** DAYS

          * SHIPS HAVE BEEN LOADED.
        THE AVERAGE IN-PORT TIME FOR A SHIP WAS  *.** HOURS.
        THE MAXIMUM IN-PORT TIME FOR A SHIP WAS  *.** HOURS.

        THE DOCK UTILIZATION WAS  *.** (X 100% )
        THE AVERAGE QUEUE FOR THE DOCK WAS  *.** SHIPS

        THE TUG UTILIZATION WAS  *.** (X 100% )
        THE AVERAGE QUEUE FOR THE TUG WAS  *.** SHIPS
    STOP
END "PROCESS GENERATOR
```

```
PROCESS SHIP
    DEFINE ARRIVAL.TIME AS A REAL VARIABLE
    DEFINE MEAN.LOAD.TIME AS A REAL VARIABLE
    IF TYPE(SHIP) = "OLD"
        LET MEAN.LOAD.TIME = LOADING.TIME
    ELSE
        LET MEAN.LOAD.TIME = 21
    ALWAYS
    UNTIL .END.OF.SIMULATION
    DO
        LET ARRIVAL.TIME = TIME.V
        REQUEST 1 DOCK(1)
        REQUEST 1 TUG(1)
        WAIT EXPONENTIAL.F(1.0,3) HOURS
        RELINQUISH 1 TUG(1)
        WORK EXPONENTIAL.F(MEAN.LOAD.TIME,4) HOURS
        REQUEST 1 TUG(1)
        WAIT EXPONENTIAL.F(1.0,5) HOURS
        RELINQUISH 1 TUG(1)
        RELINQUISH 1 DOCK(1)
        LET IN.PORT.TIME = (TIME.V - ARRIVAL.TIME) * HOURS.V
        IF TYPE(SHIP) = "OLD"
            LEAVE
        ELSE
        WAIT EXPONENTIAL.F(240.0,6) HOURS
    LOOP
END "PROCESS SHIP
```

```
                AFRICAN PORT TANKER STUDY

                NO. OF NEW SHIPS IS   5.

                        AFTER 365.02 DAYS

908 SHIPS HAVE BEEN LOADED.
  THE AVERAGE IN-PORT TIME FOR A SHIP WAS 137.69 HOURS.
  THE MAXIMUM IN-PORT TIME FOR A SHIP WAS 464.80 HOURS.

  THE DOCK UTILIZATION WAS   .95 (X 100 % )
  THE AVERAGE QUEUE FOR THE DOCK WAS   11.64 SHIPS

  THE TUG UTILIZATION WAS   .21 (X 100 % )
  THE AVERAGE QUEUE FOR THE TUG WAS   .03 SHIPS
```

FIGURE 5

```
"                SIMSCRIPT II.5 TUTORIAL MODEL
"                    AN AFRICAN PORT STUDY
PREAMBLE
    PROCESSES INCLUDE GENERATOR AND SHIP
        EVERY SHIP HAS A TYPE
            AND A SHIP.STATUS
            AND BELONGS TO THE FLEET
        DEFINE TYPE AS AN ALPHA VARIABLE
        DEFINE SHIP.STATUS AS AN ALPHA VARIABLE
        EXTERNAL PROCESSES ARE SMOG AND STORM
        EXTERNAL EVENT UNIT IS 7
    RESOURCES INCLUDE DOCK AND TUG
        AND PERMIT
    DEFINE NO.NEW.SHIPS AS AN INTEGER VARIABLE
    THE SYSTEM HAS A LOADING.TIME RANDOM STEP VARIABLE
        AND OWNS THE FLEET
    DEFINE LOADING.TIME AS A REAL VARIABLE
    ACCUMULATE AVG.DOCK.QUEUE AS THE AVERAGE OF N.Q.DOCK
    ACCUMULATE DOCK.UTILIZATION AS THE AVERAGE OF N.X.DOCK
    ACCUMULATE TUG.UTILIZATION AS THE AVERAGE OF N.X.TUG
    ACCUMULATE AVG.TUG.QUEUE AS THE AVERAGE OF N.Q.TUG
    TALLY AVG.IN.PORT.TIME AS THE AVERAGE
        AND NO.OF.SHIPS.SERVED AS THE NUMBER
        AND MAX.IN.PORT.TIME AS THE MAXIMUM OF IN.PORT.TIME
    DEFINE IN.PORT.TIME AS A REAL VARIABLE
    DEFINE .DEPART TO MEAN 1
    DEFINE .UNLOAD TO MEAN 2
    DEFINE .END.OF.SIMULATION TO MEAN TIME.V >= 365.
END "PREAMBLE

MAIN
    CALL ORIGIN.R(1,1,80)
    READ LOADING.TIME
    PRINT 1 LINE THUS
                AFRICAN PORT TANKER STUDY
    SKIP 1 LINE
    READ NO.NEW.SHIPS
    PRINT 1 LINE WITH NO.NEW.SHIPS THUS
                NO. OF NEWS SHIPS IS *.
    SKIP 1 LINE
    CREATE EVERY DOCK(1)
    LET U.DOCK(1) = 3
    CREATE EVERY TUG(1)
    LET U.TUG(1) = 1
    CREATE EVERY PERMIT(2)
    LET U.PERMIT(.UNLOAD) = 1
    LET U.PERMIT(.DEPART) = 1
    ACTIVATE A GENERATOR NOW
    START SIMULATION
END "MAIN

INPUT DATA:
    5
    0.25   18   0.55   24   0.2   36   *

PROCESS GENERATOR
    DEFINE I AS AN INTEGER VARIABLE
    FOR I = 1 TO NO.NEW.SHIPS,
    DO
        ACTIVATE A SHIP IN 2 * I DAYS
        LET TYPE(SHIP) = "NEW"
        LET SHIP.STATUS(SHIP) = "AT.SEA"
        FILE SHIP IN FLEET
    LOOP
    UNTIL .END.OF.SIMULATION,
    DO
        ACTIVATE A SHIP NOW
        LET TYPE(SHIP) = "OLD"
        LET SHIP.STATUS(SHIP) = "AT.SEA"
        FILE SHIP IN FLEET
        WAIT EXPONENTIAL.F(11.0, 2) HOURS
    LOOP
    SKIP 2 LINES
    PRINT 11 LINES WITH TIME.V, NO.OF.SHIPS,SERVED,
        AVG.IN..PORT.TIME, MAX.IN.PORT.TIME,
        DOCK.UTILIZATION(1)/3, AVG.DOCK.QUEUE(1),
        TUG.UTILIZATION(1) AND AVG.TUG.QUEUE(1) THUS
                AFTER  **.** DAYS

         * SHIPS HAVE BEEN LOADED.
        THE AVERAGE IN-PORT TIME FOR A SHIP WAS  *.** HOURS.
        THE MAXIMUM IN-PORT TIME FOR A SHIP WAS  *.** HOURS.

        THE DOCK UTILIZATION WAS  *.** (X 100% )
        THE AVERAGE QUEUE FOR THE DOCK WAS  *.** SHIPS

        THE TUG UTILIZATION WAS  *.** (X 100% )
        THE AVERAGE QUEUE FOR THE TUG WAS  *.** SHIPS
    STOP
END "PROCESS GENERATOR
```

```
PROCESS SHIP
    DEFINE ARRIVAL.TIME AS A REAL VARIABLE
    DEFINE MEAN.LOAD.TIME AS A REAL VARIABLE
    IF TYPE(SHIP) = "OLD"
        LET MEAN.LOAD.TIME = LOADING.TIME
    ELSE
        LET MEAN.LOAD.TIME = 21
    ALWAYS
    UNTIL .END.OF.SIMULATION
    DO
        LET ARRIVAL.TIME = TIME.V
        LET SHIP.STATUS(SHIP) = "IN.HARBOR"
        REQUEST 1 DOCK(1)
        REQUEST 1 TUG(1)
        WAIT EXPONENTIAL.F(1.0,3) HOURS
        RELINQUISH 1 TUG(1)
        REQUEST 1 PERMIT(.UNLOAD)
        RELINQUISH 1 PERMIT(.UNLOAD)
        LET SHIP.STATUS(SHIP) = "UNLOADING"
        WORK EXPONENTIAL.F(MEAN.LOAD.TIME,4) HOURS
        LET SHIP.STATUS(SHIP) = "IN.HARBOR"
        REQUEST 1 TUG(1)
        WAIT EXPONENTIAL.F(1.0,5) HOURS
        RELINQUISH 1 TUG(1)
        RELINQUISH 1 DOCK(1)
        REQUEST 1 PERMIT(.DEPART)
        RELINQUISH 1 PERMIT(.DEPART)
        LET IN.PORT.TIME = (TIME.V - ARRIVAL.TIME) * HOURS V
        IF TYPE(SHIP) = "OLD"
            REMOVE THIS SHIP FROM FLEET
            LEAVE
        ELSE
        LET SHIP.STATUS(SHIP) = "AT.SEA"
        WAIT EXPONENTIAL.F(240.0,6) HOURS
    LOOP
END "PROCESS SHIP

PROCESS STORM
    DEFINE DURATION AS A REAL VARIABLE
    READ DURATION
    REQUEST 1 PERMIT(.DEPART) WITH PRIORITY 1
    FOR EACH SHIP IN FLEET,
        WITH SHIP.STATUS(SHIP) = "AT.SEA"
        INTERRUPT SHIP
    WAIT DURATION DAYS
    RELINQUISH 1 PERMIT(.DEPART)
    FOR EACH SHIP IN FLEET,
        WITH SHIP.STATUS(SHIP) = "AT.SEA"
        RESUME SHIP
END "STORM

PROCESS SMOG
    DEFINE DURATION AS A REAL VARIABLE
    READ DURATION
    REQUEST 1 PERMIT(.UNLOAD) WITH PRIORITY 1
    FOR EACH SHIP IN FLEET,
        WITH SHIP.STATUS(SHIP) = "UNLOADING"
        INTERRUPT SHIP
    WAIT DURATION DAYS
    RELINQUISH 1 PERMIT(.UNLOAD)
    FOR EACH SHIP IN FLEET,
        WITH SHIP.STATUS(SHIP) = "UNLOADING"
        RESUME SHIP
END "SMOG
```

```
                AFRICAN PORT TANKER STUDY

                NO. OF NEW SHIPS IS   5.

                        AFTER 365.02 DAYS

928 SHIPS HAVE BEEN LOADED.
  THE AVERAGE IN-PORT TIME FOR A SHIP WAS 112.49 HOURS.
  THE MAXIMUM IN-PORT TIME FOR A SHIP WAS 461.69 HOURS.

  THE DOCK UTILIZATION WAS    .95 (X 100 % )
  THE AVERAGE QUEUE FOR THE DOCK WAS   9.12 SHIPS

  THE TUG UTILIZATION WAS   .22 (X 100 % )
  THE AVERAGE QUEUE FOR THE TUG WAS   .04 SHIPS
```

FIGURE 6

```
"                SIMSCRIPT II.5 TUTORIAL MODEL
"                    AN AFRICAN PORT STUDY
PREAMBLE
    PROCESSES INCLUDE GENERATOR AND SHIP
        EVERY SHIP HAS A TYPE
            AND A SHIP.STATUS
            AND A SHIP.ID
            AND BELONGS TO THE FLEET
        DEFINE TYPE AS AN ALPHA VARIABLE
        DEFINE SHIP.STATUS AS AN ALPHA VARIABLE
        DEFINE SHIP.ID AS AN INTEGER VARIABLE
        EXTERNAL PROCESSES ARE SMOG AND STORM
        EXTERNAL EVENT UNIT IS 7
    RESOURCES INCLUDE DOCK AND TUG
        AND PERMIT
    DEFINE NO.NEW.SHIPS AS AN INTEGER VARIABLE
    THE SYSTEM HAS A LOADING.TIME RANDOM.STEP VARIABLE
        AND OWNS THE FLEET
    DEFINE LOADING.TIME AS A REAL VARIABLE
    ACCUMULATE AVG.DOCK.QUEUE AS THE AVERAGE OF N.Q.DOCK
    ACCUMULATE DOCK.UTILIZATION AS THE AVERAGE OF N.X.DOCK
    ACCUMULATE TUG.UTILIZATION AS THE AVERAGE OF N.X.TUG
    ACCUMULATE AVG.TUG.QUEUE AS THE AVERAGE OF N.Q.TUG
    TALLY AVG.IN.PORT.TIME AS THE AVERAGE
        AND NO.OF.SHIPS.SERVED AS THE NUMBER
        AND MAX.IN.PORT.TIME AS THE MAXIMUM OF IN.PORT.TIME
    DEFINE IN.PORT.TIME AS A REAL VARIABLE
    DEFINE .DEPART TO MEAN 1
    DEFINE .UNLOAD TO MEAN 2
    DEFINE .END.OF.SIMULATION TO MEAN TIME.V >= 365.
END "PREAMBLE

MAIN
    LET BETWEEN.V = 'DETAIL'
    CALL ORIGIN.R(1,1,80)
    READ TYPE.DISTRIBUTION
    PRINT 1 LINE THUS
                AFRICAN PORT TANKER STUDY
    SKIP 1 LINE
    READ NO.NEW.SHIPS
    PRINT 1 LINE WITH NO.NEW.SHIPS THUS
                NO. OF NEWS SHIPS IS *.
    SKIP 1 LINE
    CREATE EVERY DOCK(1)
    LET U.DOCK(1) = 3
    CREATE EVERY TUG(1)
    LET U.TUG(1) = 1
    CREATE EVERY PERMIT(2)
    LET U.PERMIT(.UNLOAD) = 1
    LET U.PERMIT(.DEPART) = 1
    ACTIVATE A GENERATOR NOW
    START SIMULATION
END "MAIN

INPUT DATA:
    5
    0.25  18  0.55  24  0.2  36  *

PROCESS GENERATOR
    DEFINE I AS AN INTEGER VARIABLE
    DEFINE SHIP.COUNT AS AN INTEGER, SAVED VARIABLE
    FOR I = 1 TO NO.NEW.SHIPS,
    DO
        ACTIVATE A SHIP IN 2 * I DAYS
        LET TYPE(SHIP) = "NEW"
        LET SHIP.STATUS(SHIP) = "AT.SEA"
        ADD 1 TO SHIP.COUNT
        LET SHIP.ID(SHIP) = SHIP.COUNT
        FILE SHIP IN FLEET
    LOOP
    UNTIL .END.OF.SIMULATION,
    DO
        ACTIVATE A SHIP NOW
        LET TYPE(SHIP) = "OLD"
        LET SHIP.STATUS(SHIP) = "AT.SEA"
        ADD 1 TO SHIP.COUNT
        LET SHIP.ID(SHIP) = SHIP.COUNT
        FILE SHIP IN FLEET
        WAIT EXPONENTIAL.F(11.0, 2) HOURS
    LOOP
    SKIP 2 LINES
    PRINT 11 LINES WITH TIME.V, NO.OF.SHIPS.SERVED,
        AVG.IN..PORT.TIME, MAX.IN.PORT.TIME,
        DOCK.UTILIZATION(1)/3, AVG.DOCK.QUEUE(1),
        TUG.UTILIZATION(1) AND AVG.TUG.QUEUE(1) THUS
                AFTER **.** DAYS

         * SHIPS HAVE BEEN LOADED.
        THE AVERAGE IN-PORT TIME FOR A SHIP WAS  *.** HOURS.
        THE MAXIMUM IN-PORT TIME FOR A SHIP WAS  *.** HOURS.

        THE DOCK UTILIZATION WAS  *.** (X 100% )
        THE AVERAGE QUEUE FOR THE DOCK WAS  *.** SHIPS

        THE TUG UTILIZATION WAS  *.** (X 100% )
        THE AVERAGE QUEUE FOR THE TUG WAS  *.** SHIPS
    STOP
END "PROCESS GENERATOR

PROCESS SHIP
    DEFINE ARRIVAL.TIME AS A REAL VARIABLE
    DEFINE MEAN.LOAD.TIME AS A REAL VARIABLE
    IF TYPE(SHIP) = "OLD"
        LET MEAN.LOAD.TIME = LOADING.TIME
    ELSE
        LET MEAN.LOAD.TIME = 21
    ALWAYS
    UNTIL .END.OF.SIMULATION
    DO
        LET ARRIVAL.TIME = TIME.V
        LET SHIP.STATUS(SHIP) = "IN.HARBOR"
        REQUEST 1 DOCK(1)
        REQUEST 1 TUG(1)
        WAIT EXPONENTIAL.F(1.0,3) HOURS
        RELINQUISH 1 TUG(1)
        REQUEST 1 PERMIT(.UNLOAD)
        RELINQUISH 1 PERMIT(.UNLOAD)
        LET SHIP.STATUS(SHIP) = "UNLOADING"
        WORK EXPONENTIAL.F(MEAN.LOAD.TIME,4) HOURS
        LET SHIP.STATUS(SHIP) = "IN.HARBOR"
        REQUEST 1 TUG(1)
        WAIT EXPONENTIAL.F(1.0,5) HOURS
        RELINQUISH 1 TUG(1)
        RELINQUISH 1 DOCK(1)
        REQUEST 1 PERMIT(.DEPART)
        RELINQUISH 1 PERMIT(.DEPART)
        LET IN.PORT.TIME = (TIME.V - ARRIVAL.TIME) * HOURS V
        IF TYPE(SHIP) = "OLD"
            REMOVE THIS SHIP FROM FLEET
            LEAVE
        ELSE
        LET SHIP.STATUS(SHIP) = "AT.SEA"
        WAIT EXPONENTIAL.F(240.0,6) HOURS
    LOOP
END "PROCESS SHIP

PROCESS STORM
    DEFINE DURATION AS A REAL VARIABLE
    READ DURATION
    REQUEST 1 PERMIT(.DEPART) WITH PRIORITY 1
    FOR EACH SHIP IN FLEET,
        WITH SHIP.STATUS(SHIP) = "AT.SEA"
        INTERRUPT SHIP
    WAIT DURATION DAYS
    RELINQUISH 1 PERMIT(.DEPART)
    FOR EACH SHIP IN FLEET,
        WITH SHIP.STATUS(SHIP) = "AT.SEA"
        RESUME SHIP
END "STORM

PROCESS SMOG
    DEFINE DURATION AS A REAL VARIABLE
    READ DURATION
    REQUEST 1 PERMIT(.UNLOAD) WITH PRIORITY 1
    FOR EACH SHIP IN FLEET,
        WITH SHIP.STATUS(SHIP) = "UNLOADING"
        INTERRUPT SHIP
    WAIT DURATION DAYS
    RELINQUISH 1 PERMIT(.UNLOAD)
    FOR EACH SHIP IN FLEET,
        WITH SHIP.STATUS(SHIP) = "UNLOADING"
        RESUME SHIP
END "SMOG

ROUTINE DETAIL
    DEFINE LINE.COUNT AS AN INTEGER, SAVED VARIABLE
    ADD 1 TO LINE.COUNT
    IF LINE.COUNT = 25,
        LET BETWEEN.V = 0
    ALWAYS
    IF EVENT.V = I.SHIP
        PRINT 1 LINE WITH TIME.V, N.Q.DOCK(1), N.Q.TUG(1), N.X.DOCK(1),
            N.X.TUG(1), U.DOCK(1) THUS
AT TIME  ***.**  N.Q.DOCK=  * N.Q.TUG=  * IN-DOCK=  * IN-TUG=  * FREE DOCKS *
        PRINT 1 LINE WITH SHIP.ID(SHIP), TYPE(SHIP) AND LINE.F(SHIP) THUS
            SHIP  ****** (TYPE ***) WILL RESUME AT LINE  * .
        RETURN
    OTHERWISE
    IF EVENT.V = I.SMOG,
        PRINT 1 LINE WITH LINE.F(SMOG) AND TIME.V THUS
SMOG PROCESS RESUMING ON LINE  * AT TIME  ***.**.
        RETURN
    OTHERWISE
    IF EVENT.V = I.STORM,
        PRINT 1 LINE WITH LINE.F(STORM) AND TIME.V THUS
STORM PROCESS RESUMING ON LINE  * AT TIME  ***.**.
        RETURN
    OTHERWISE
END "ROUTINE DETAIL
```

FIGURE 7

```
ROUTINE FOR SNAP.R
    START NEW PAGE
    PRINT 1 LINE THUS
    GLOBAL CONDITIONS AT SNAP
    SKIP 2 LINES
    FOR EACH SHIP IN THE FLEET,
        PRINT 1 LINE WITH SHIP.ID(SHIP) AND TYPE(SHIP) AND LINE.F(SHIP) THUS
    SHIP ****** (TYPE  ***) IS AWAITING EXECUTION OF LINE   * .
    SKIP 2 LINES
    PRINT 3 LINES WITH U.DOCK(1), N.X.DOCK(1) AND N.Q.DOCK(1) THUS
    THERE ARE   * FREE DOCKS
    THERE ARE   * DOCKS IN USE
    AND   * SHIPS IN THE DOCK QUEUE.
    SKIP 2 LINES
    PRINT 3 LINES WITH U.TUG(1), N.X.TUG(1) AND N.Q.TUG(1) THUS
    THERE ARE   * FREE TUGS
    THERE ARE   * TUGS IN USE
    AND   * SHIPS IN THE TUG QUEUE.
    SKIP 2 LINES
END "SNAP.R
```

```
                AFRICAN PORT TANKER STUDY
                NO. OF NEW SHIPS IS   5.
AT TIME     0.  N.Q.DOCK=  0 N.Q.TUG=  0 IN-DOCK=  0 IN-TUG=  0 FREE DOCKS  3
            SHIP      6 (TYPE OLD) WILL RESUME AT LINE   1 .
AT TIME     .03 N.Q.DOCK=  0 N.Q.TUG=  0 IN-DOCK=  1 IN-TUG=  0 FREE DOCKS  2
            SHIP      6 (TYPE OLD) WILL RESUME AT LINE 16 .
STORM PROCESS RESUMING ON LINE  1 AT TIME    .33.
SMOG PROCESS RESUMING ON LINE  1 AT TIME    .50.
AT TIME     .77 N.Q.DOCK=  0 N.Q.TUG=  0 IN-DOCK=  1 IN-TUG=  0 FREE DOCKS  2
            SHIP      7 (TYPE OLD) WILL RESUME AT LINE   1 .
AT TIME     .81 N.Q.DOCK=  0 N.Q.TUG=  0 IN-DOCK=  2 IN-TUG=  1 FREE DOCKS  1
            SHIP      7 (TYPE OLD) WILL RESUME AT LINE 16 .
SMOG PROCESS RESUMING ON LINE  9 AT TIME   1.00.
AT TIME    1.00 N.Q.DOCK=  0 N.Q.TUG=  0 IN-DOCK=  2 IN-TUG=  0 FREE DOCKS  1
            SHIP      7 (TYPE OLD) WILL RESUME AT LINE 18 .
AT TIME    1.23 N.Q.DOCK=  0 N.Q.TUG=  0 IN-DOCK=  2 IN-TUG=  0 FREE DOCKS  1
            SHIP      7 (TYPE OLD) WILL RESUME AT LINE 21 .
AT TIME    1.24 N.Q.DOCK=  0 N.Q.TUG=  0 IN-DOCK=  2 IN-TUG=  1 FREE DOCKS  1
            SHIP      7 (TYPE OLD) WILL RESUME AT LINE 24 .
AT TIME    1.55 N.Q.DOCK=  0 N.Q.TUG=  0 IN-DOCK=  1 IN-TUG=  0 FREE DOCKS  2
            SHIP      8 (TYPE OLD) WILL RESUME AT LINE   1 .
AT TIME    1.56 N.Q.DOCK=  0 N.Q.TUG=  0 IN-DOCK=  2 IN-TUG=  1 FREE DOCKS  1
            SHIP      8 (TYPE OLD) WILL RESUME AT LINE 16 .
AT TIME    1.61 N.Q.DOCK=  0 N.Q.TUG=  0 IN-DOCK=  2 IN-TUG=  0 FREE DOCKS  1
            SHIP      9 (TYPE OLD) WILL RESUME AT LINE   1 .
AT TIME    1.64 N.Q.DOCK=  0 N.Q.TUG=  0 IN-DOCK=  3 IN-TUG=  1 FREE DOCKS  0
            SHIP      9 (TYPE OLD) WILL RESUME AT LINE 16 .
AT TIME    1.81 N.Q.DOCK=  0 N.Q.TUG=  0 IN-DOCK=  3 IN-TUG=  0 FREE DOCKS  0
            SHIP      9 (TYPE OLD) WILL RESUME AT LINE 21 .
AT TIME    1.84 N.Q.DOCK=  0 N.Q.TUG=  0 IN-DOCK=  3 IN-TUG=  1 FREE DOCKS  0
            SHIP      9 (TYPE OLD) WILL RESUME AT LINE 24 .
AT TIME    2.26 N.Q.DOCK=  0 N.Q.TUG=  0 IN-DOCK=  2 IN-TUG=  0 FREE DOCKS  1
            SHIP      8 (TYPE OLD) WILL RESUME AT LINE 21 .
AT TIME    2.26 N.Q.DOCK=  0 N.Q.TUG=  0 IN-DOCK=  2 IN-TUG=  1 FREE DOCKS  1
            SHIP      8 (TYPE OLD) WILL RESUME AT LINE 24 .
STORM PROCESS RESUMING ON LINE  9 AT TIME   2.33.
AT TIME    2.33 N.Q.DOCK=  0 N.Q.TUG=  0 IN-DOCK=  1 IN-TUG=  0 FREE DOCKS  2
            SHIP      7 (TYPE OLD) WILL RESUME AT LINE 27 .
AT TIME    2.33 N.Q.DOCK=  0 N.Q.TUG=  0 IN-DOCK=  1 IN-TUG=  0 FREE DOCKS  2
            SHIP      9 (TYPE OLD) WILL RESUME AT LINE 27 .


                        AFTER 365.02 DAYS
928 SHIPS HAVE BEEN LOADED.
  THE AVERAGE IN-PORT TIME FOR A SHIP WAS 112.49 HOURS.
  THE MAXIMUM IN-PORT TIME FOR A SHIP WAS 461.69 HOURS.

  THE DOCK UTILIZATION WAS    .95 (X 100 % )
  THE AVERAGE QUEUE FOR THE DOCK WAS   9.12 SHIPS

  THE TUG UTILIZATION WAS   .22 (X 100 % )
  THE AVERAGE QUEUE FOR THE TUG WAS   .04 SHIPS
```

FIGURE 7
(Continued)